```
model
        {
         for (i in 1 : n) {
          #
          #model for aspirin treatment - logistic regression on centre & nice
          #
          logit(pa[i])<- a1*nice[i]+centre.effect.a[centre[i]]
          aspirin[i]~dbin(pa[i],1)
          #
          #model for true events - logistic regression on centre and nice
          #
          logit(py[i])<- b1*nice[i]+centre.effect.y[centre[i]]
          y[i]~dbin(py[i],1)
          #
          # model for observed events
          #
            p.events[i]<-y[i]*(1-0.3*aspirin[i])
            events[i]~ dbin(p.events[i],1)
            y1[i]<-y[i]*nice[i]*risk[i]
            y2[i]<-y[i]*(1-nice[i])*risk[i]
            y3[i]<-y[i]*nice[i]*(1-risk[i])
            y4[i]<-y[i]*(1-nice[i])*(1-risk[i])
                        }
        #
        # Prior specification
        #
        for ( j in 1:n.centre ){
        centre.effect.a[j]~dnorm(0,0.001)
        centre.effect.y[j]~dnorm(0,0.001)
        }
        a1~dnorm(0,0.0001)
        b1~dnorm(0,0.0001)
        #
        # True event counts
        #
        s1<-sum(y1[]) # NICE + Risk +
        s2<-sum(y2[]) # NICE  - Risk +
        s3<-sum(y3[]) # NICE + Risk -
        s4<-sum(y4[]) # NICE  - Risk -
                            }
```